
11

Testing UDFs for
Correctness and

Performance

The first section of this chapter is omitted.  It deals with the issue
of testing UDFs for correctness.



Drilling Down into the Performance Problem

The problem of inadequate performance is the primary reason not to use
scalar UDFs. There can be a big difference in the performance of a query
that employs UDFs and one that replaces them with equivalent
expressions.

Using UDFs gives your query the same performance as if you imple-
mented the query using a cursor instead of relational SQL. That’s because
the SQL Server query engine has to address every row and invoke the
UDF. It creates a loop very much like a cursor.

Using UDFs is great! Most of the time. But there are times when the
extra overhead of UDF execution is more than you can tolerate. Elimi-
nating the UDF, where possible, is a strategy that usually produces
dramatically faster query execution times. By “dramatically” I mean two
to 500 times faster (enough to make a real difference in the response time
of your application).

This section demonstrates and documents the performance problem
in detail by constructing an experiment that compares the performance of
a UDF-based solution with one that replaces the UDF with an expression.
The first step is to pick a suitable UDF for the experiment—one that can
be replaced by an expression. Secondly, to make the results dramatic
enough, we’ll build a table with a sufficient number of rows to show the
difference between the UDF and the expression. Finally, we’ll run the
experiment.

The Function and the Template for Testing
Performance

The function that we’re going to use is udf_Txt_CharIndexRev, which is
shown in Listing 11.2. It works just like the built-in CHARINDEX function
except that it works from the back of the string. Like CHARINDEX, the posi-
tion it reports is relative to the beginning of the string that’s searched. I
use udf_Txt_CharIndexRev for tasks like finding the slash that separates a
file name from the rest of a file path.

Listing 11.2: udf_Txt_CharIndexRev

CREATE FUNCTION dbo.udf_Txt_CharIndexRev (

@SearchFor varchar(255) -- Sequence to be found
, @SearchIn varchar(8000) -- The string to be searched

) RETURNS int -- Position from the back of the string where
-- @SearchFor is found in @SearchIn

WITH SCHEMABINDING
/*
* Searches for a string in another string working from the back.
* It reports the position (relative to the front) of the first
* such expression it finds. If the expression is not found, it
* returns zero.
*
* Equivalent Template:

CASE
WHEN CHARINDEX(<search_for, varchar(255), ''>

, <search_in, varchar(8000), ''>) > 0
THEN LEN(<search_in, varchar(8000), ''>)

- CHARINDEX(REVERSE(<search_for, varchar(255), ''>)
, REVERSE (<search_in, varchar(8000), ''>)) + 1

ELSE 0 END
* Example:
select dbo.udf_Txt_CharIndexRev('\', 'C:\temp\abcd.txt')
*
* Test:
PRINT 'Test 1 ' + CASE WHEN 8=

dbo.udf_Txt_CharIndexRev ('\', 'C:\temp\abcd.txt')
THEN 'Worked' ELSE 'ERROR' END

PRINT 'Test 2 ' + CASE WHEN 0=
dbo.udf_TxtCharIndexRev ('*', 'C:\tmp\d.txt')

THEN 'Worked' ELSE 'ERROR' END
****************************************************************/
AS BEGIN

DECLARE @Result int
, @StringLen int
, @ReverseIn varchar(8000)
, @ReverseFor varchar(255)

SELECT @ReverseIn = REVERSE (@SearchIn)
, @ReverseFor = REVERSE(@SearchFor)
, @StringLen = LEN(@SearchIn)

SELECT @Result = CHARINDEX(@ReverseFor, @ReverseIn)

-- return the position from the front of the string
IF @Result > 0

SET @Result = @StringLen - @Result + 1
-- ENDIF

RETURN @Result
END

Before we get to the experiment, let’s try out a few simple cases with this
test query:

-- Demonstrate udf_Txt_CharIndexRev
SELECT dbo.udf_Txt_CharIndexRev('fdf', 'f123 asdasfdfdfddfjas ')

as [Middle]
, dbo.udf_Txt_CharIndexRev('C', 'C:\temp\ab.txt') as [start]
, dbo.udf_Txt_CharIndexRev('X', '123456789X') as [end]
, dbo.udf_Txt_CharIndexRev('AB', '12347') as [missing]

GO

(Results)

Middle start end missing
----------- ----------- ----------- -----------

13 1 10 0

udf_Txt_CharIndexRev is simple enough that it’s possible to replace it with
an expression. The section in the comment block labeled “Equivalent
Template” tells how to achieve the same result as the UDF without using
the UDF. Here’s the template:

CASE
WHEN CHARINDEX(<search_for, varchar(255), ''>

, <search_in, varchar(8000), ''>) > 0
THEN LEN(<search_in, varchar(8000), ''>)

- CHARINDEX(REVERSE(<search_for, varchar(255), ''>)
, REVERSE (<search_in, varchar(8000), ''>)) + 1

ELSE 0 END

Why bother having the UDF at all if it can be replaced with an expression?
As you may recall, my philosophy about writing efficient code is that cod-
ing is an economic activity with trade-offs between the cost of writing
code and the cost of running it. In my opinion, the best course of action is
to write good, easy-to-maintain code and use a basic concern for perfor-
mance to eliminate any obvious performance problem. The overhead of
using a UDF to parse the extension from one file name isn’t ever going to
show up on the performance radar screen. It’s when a UDF is used on
large numbers of rows that it becomes a problem. In situations with few
rows, having the UDF helps by making the code easier to write.

The Equivalent Template section of the comment block is used to
replace a function call to udf_Txt_CharIndexRev with the text in the tem-
plate and then, using SQL Query Analyzer’s menu item Edit � Replace
Template Parameters, put the function parameters into the revised SQL at
the locations where they belong. The Query Analyzer template facility
comes in handy because the parameters have to be inserted in more than
one place.

So that we don’t have to perform the replacement process four times,
let’s convert this query that does just one call to udf_Txt_CharIndexRev
into an equivalent query using the template:

-- Query to translate
SELECT dbo.udf_Txt_CharIndexRev('fad', 'f123 asdasfdfdfddfjas ')

as [Middle]
GO

The first step is to replace the function call with the Equivalent Template,
and we get:

SELECT CASE
WHEN CHARINDEX(<search_for, varchar(255), ''>

, <search_in, varchar(8000), ''>) > 0
THEN LEN(<search_in, varchar(8000), ''>)

- CHARINDEX(REVERSE(<search_for, varchar(255), ''>)
, REVERSE (<search_in, varchar(8000), ''>)) + 1

ELSE 0 END
as [Middle]

Next we use the Edit � Replace Template Parameters menu item and
enter the two parameters. Figure 11.2 shows the Replace Template
Parameters dialog box with the replacement text entered.

After pressing the Replace All button, the result looks like this:

-- Translated query
SELECT CASE

WHEN CHARINDEX('fdf'
, 'f123 asdasfdfdfddfjas ') > 0

THEN LEN('f123 asdasfdfdfddfjas ')
- CHARINDEX(REVERSE('fad')

, REVERSE ('f123 asdasfdfdfddfjas ')) + 1
ELSE 0 END

as [Middle]
GO

(Results)

Middle
-----------

13

As you can see, replacing the call to the UDF with the equivalent expres-
sion results in a much longer and messier query. Simplification, with a
corresponding reduction in maintenance effort, is a benefit of using UDFs
that is lost when you replace them with complex expressions.

Now that we have a function and can replace it with an expression, we
need some data that helps show us the difference in performance between
the two. Pubs and Northwind are useful for learning but they don’t contain
very much data; at best they have a few hundred rows in any one table. I
like to run tests on a million-row table. SQL Server doesn’t come with any
tables that large, so we’ll have to construct one.

Constructing a Large Test Table
To test the difference between the UDF and the equivalent expression, we
need a table with a large number of rows. Figure 11.3 shows a database
diagram for the ExampleNumberString table that will be used for the experi-
ment. The NumberString column is a 20- to 23-digit numeric string that we
can use for experimentation on udf_Txt_CharIndexRev.

The table hasn’t been added to your TSQLUDFS database because there
is a stored procedure in the database to create the table and populate it
with plenty of rows. usp_CreateExampleNumberString is shown in Listing

11.3. Its parameter is the number of times to double the number of rows
in ExampleNumberString.

Listing 11.3: usp_CreateExampleNumberString

CREATE PROC usp_CreateExampleNumberString

@Loops int = 20 -- creates POWER (2, @Loops) Rows
-- 20 Creates 1,000,000 rows

AS

DECLARE @LC int -- Loop counter

-- delete an existing ExampleNumberString table
if exists (select * from dbo.sysobjects

where id = object_id(N'dbo.ExampleNumberString')
and OBJECTPROPERTY(id, N'IsUserTable') = 1)

DROP TABLE dbo.ExampleNumberString

CREATE TABLE ExampleNumberString (
ID int identity (1,1)

, BigNum Numeric (38, 0)
, NumberString varchar(128) NULL
)

INSERT INTO ExampleNumberString (BigNum, NumberString)
VALUES(CONVERT (numeric(38,0), rand() * 9999999999999)

, ' ') -- preallocate

SELECT @LC = 0
WHILE @LC < @Loops BEGIN

INSERT INTO ExampleNumberString (BigNum, NumberString)
SELECT BigNum * RAND(@LC + 1) * 2

, ' '
FROM ExampleNumberString

SELECT @LC = @LC + 1
END -- WHILE

UPDATE ExampleNumberString
SET NumberString = convert(varchar(128)

, convert(numeric(38,0), 9834311) * bignum)

Before running usp_CreateExampleNumberString, check to be sure you have
enough disk space. Adding one million rows takes about 70 megabytes.
With the size of today’s disks, that shouldn’t be a problem, but check any-
way. To add the million rows, run the query as is. If you can’t spare the 70
megabytes, reduce the parameter from 20 to just 10, which will add only
about 1,000 rows. Now create the table:

-- Create and populate the table
DECLARE @RC int
EXEC @RC = usp_CreateExampleNumberString 20
GO

usp_CreateExampleNumberString does a reasonable job of producing a differ-
ent large NumberString for each row. There is usually about one percent
duplication of numbers, which seems acceptable for most testing scenar-
ios. With a parameter of 20, the procedure takes about 55 seconds to run
on my desktop development system.

Since we’re going to run multiple queries against the NumberString
column to test the time it takes to use a UDF on a large number of rows,
we should try to make the circumstances for all queries as similar as pos-
sible. One way to do that is to force all the rows that are going to be
queried into memory, SQL Server’s page cache, and keep them there. If
we don’t have all the desired rows in memory, then we might be compar-
ing one query that reads a million rows from disk with another query that
reads the same million rows from SQL Server’s page cache. It wouldn’t be
a fair comparison.

SQL Server provides the DBCC PINTABLE statement to force all pages
from a table to remain in the cache once they’ve been read the first time.
Use it with caution! It can fill SQL Server’s cache and cause the query
engine to lock up to the point where you have to shut it down and restart
it. Only do this if you have adequate RAM available, not just virtual mem-
ory. There’s no point in using virtual memory as a substitute for RAM in
this situation. That just substitutes one form of disk I/O (paging) for the
one we’re trying to eliminate (reading pages from disk).

This next script pins ExampleNumberString in memory. On my desktop
development system with a million-row table, pinning the table forced
SQL Server to consume 82 additional megabytes of RAM. The system has
512 megabytes of RAM, and I used Task Manager to see that memory
isn’t full. If your system has less available RAM, reduce the number of
rows in the test to eliminate paging.

-- Pin ExampleNumberString into memory. You must change the database name, if
-- you're not going to run it in TSQLUDFS
-- Be sure that you have enough memory available before you do
-- this. My SQL Server Process grew to 82 megabytes when I ran
-- this script.
DECLARE @db_id int, @tbl_id int
SET @db_id = DB_ID('TSQLUDFS') -- DB_ID('<your database name goes here>')
SET @tbl_id = OBJECT_ID('ExampleNumberString')
DBCC PINTABLE (@db_id, @tbl_id)
GO

(Results)

Warning: Pinning tables should be carefully considered. If a pinned table is larger,
or grows larger, than the available data cache, the server may need to be restarted
and the table unpinned.
DBCC execution completed. If DBCC printed error messages, contact your system
administrator.

Pinning the table only tells SQL Server to never remove the table’s pages
from the page cache. It doesn’t read them into the cache. The next query
does:

-- Read all the rows to force the pages into the cache
SELECT * from ExampleNumberString
GO

With a million-row table in memory, the stage is set for comparing the
UDF and the equivalent template. Ladies and gentleman, place your bets.

Experimenting with UDF vs. Expression Performance
Let’s start by constructing a query that uses udf_Txt_CharIndexRev to
search for the string '83' from the back of the NumberString column and
then replace the UDF with the equivalent expression. The two queries
are:

SELECT dbo.udf_Txt_CharIndexRev ('83', NumberString)
as [Index of 83 from the string end]

FROM ExampleNumberString

SELECT CASE WHEN CHARINDEX('83', NumberString) > 0
THEN LEN(NumberString)

- CHARINDEX(REVERSE('83')
, REVERSE (NumberString)) + 1

ELSE 0 END as [Index of 83 from the string end]
FROM ExampleNumberString

Don’t run them yet. I have a few more wrinkles to throw into the
experiment.

Any query that returns a million rows to SQL Query Analyzer’s
results window is going to do a lot of work on sending, receiving, and dis-
playing the results. To eliminate most of that work, using an aggregate
function, such as MAX, forces the UDF or expression to be evaluated with-
out returning much data as the result.

For comparison purposes, I’ve also included a third query, labeled
Query #0, that just takes the MAX of the LEN function. This query repre-
sents the minimum time it might take to just read the million rows of data.

The SET STATISTICS TIME ON command tells SQL Server to measure
the time to parse and compile the query and the time to execute the query
and report them back with the query results.

Making these three changes to the queries gives us our experiment.
If you run them, be patient. Query #1 took over four minutes on my sys-
tem. The three queries with their results from my desktop development
system are:

-- Query #0: The minimum time for an operation that scans the whole table.
SET STATISTICS TIME ON
SELECT MAX(LEN(NumberString)) as [Max Len]

FROM ExampleNumberString
GO

(Results)

SQL Server parse and compile time:
CPU time = 2 ms, elapsed time = 2 ms.Max Length

-----------
23

SQL Server Execution Times:
CPU time = 991 ms, elapsed time = 991 ms.

(End of Results for Query #0)

-- Query #1: The UDF. Note execution time is incorrect.
SET STATISTICS TIME ON
SELECT MAX (dbo.udf_Txt_CharIndexRev ('83', NumberString))

as [Right Most Position]
FROM ExampleNumberString

GO

(Results)

SQL Server parse and compile time:
CPU time = 2 ms, elapsed time = 2 ms.

Right Most Position
-------------------

23

SQL Server Execution Times:
CPU time = 0 ms, elapsed time = 0 ms.

(End of Results for Query #1 - It took 4 minutes 38 seconds)
( - CPU was working at 100% during that time)

-- Query #2: The equivalent expression
SET STATISTICS TIME ON
SELECT MAX (CASE WHEN CHARINDEX('83', NumberString) > 0

THEN LEN(NumberString) - CHARINDEX(REVERSE('83')
, REVERSE (NumberString)) + 1

ELSE 0 END
) [Right Most Position]

FROM ExampleNumberString
GO

(Results)

SQL Server parse and compile time:
CPU time = 3 ms, elapsed time = 3 ms.

Right Most Position
-------------------

23

SQL Server Execution Times:
CPU time = 2934 ms, elapsed time = 2954 ms.

(End of Results for Query #2)

Query #1, the UDF, didn’t report its execution times correctly. It seems
that SET STATISTICS TIME is limited in the duration that it can measure. I
used the Windows Task Manager to watch what was happening on my sys-
tem, and the CPU time for Query #1 is very close to the elapsed time. For
our comparison, we’ll have to use the elapsed time.

Table 11.1 has the comparison in run time of the queries. The Net col-
umn subtracts the time it took to run Query #0, the very simplest
expression on the same set of strings, from the other queries. I think that
the Net column has the numbers that should be compared because it iso-
lates just the effect of running the UDF or the replacement expression.

Table 11.1: Query timings for the experiment

Query Description Time (milliseconds) Net

#0 Simple expression 991 N/A

#1 UDF 278000 277009

#2 UDF replaced by expression 2954 1963

The difference in time is dramatic. The UDF takes about 140 times longer
to run as an equivalent expression. Wow! Four-plus minutes versus three
seconds is the kind of difference users really notice.

Of course, the difference isn’t going to be perceivable when the query
is run on 1, 5, or even 100 rows. Only when the number of rows grows
into the thousands does the difference begin to be noticeable.

The lesson that I draw from this experiment is that UDFs must be
used with care in performance-sensitive situations. They’re a great tool
for simplifying code and promoting code reuse, but they can have a dra-
matically negative effect on performance.

